An efficient and exact solution for Random Vibration Analysis using MSC/NASTRAN. Part I: White noise spectrum

E. de la Fuente and J. San Millan
Instituto Nacional de Tecnica Aeroespacial
Structures and Mechanisms Department
28850 Torrejon de Ardoz. Madrid, Spain

Abstract

A new method for performing RANDOM vibration analysis within MSC/NASTRAN is presented in this paper. The method is a direct application of a well known result of Linear Systems Theory and allows *exact* computation of RMS values of any number of structural responses and they can be postprocessed as if they were originated in a conventional static analysis (in colour plots for instance). Also the DMAP sequence that allows to use it within MSC/NASTRAN solver is included and described. The comparison of the obtained results with those given by standard method shows the correctness of the DMAP sequence. Finally, extensions of the capability of the presented method are outlined.

1 Introduction

RANDOM vibration analysis is performed within MSC/NASTRAN in Frequency Response solution (SOL 111). This RANDOM solution postproccesses the transfer functions of the results requested by the user, i.e. stresses, forces, accelerations, etc., by means of numerical integration of the PSD curves of the responses, calculated from the corresponding ones of the excitation forces and the frequency response of the structure. The calculation process presents some difficulties very well known by the users:

- The results have to be requested one by one (no instruction as STRESS=ALL is available). This inconvenient is important when dealing with big structures
- An relatively important number of mathematical operations have to be made for the calculation of each individual result.
- It is not possible to visualize RMS results (stresses for instance) in color plots as it is made in static or modal analysis, because RANDOM results are not stored in the OUTPUT2 file produced by MSC/NASTRAN.. This is obviously important to have a clear idea of the part of the structure that is most loaded.
- The solution obtained is not exact because the results PSD are calculated by means of a numeric integration

The method presented in this paper tries to solve these shortcomings. The main advantage with respect to the sequence already implemented within MSC/NASTRAN solution 111, is that all stresses, forces or any result requested by the user may be computed, and it is possible to postprocess the RMS results as if they came from a conventional static analysis (color plots can be obtained showing RMS stresses, displacements and so on). Other are the exactness and higher efficiency (the calculation sequence is clearly simpler and shorter that the default one).

Obviously, there is an associated cost however. The main limitation of the proposed method is that it is only valid for uniform input RANDOM spectra (i.e. white noise one). Nevertheless this limitation can be easily overcome in most practical cases, as it will be described in a future paper.

A DMAP ALTER (also included in the paper) was developed to implement the method in MSC/NASTRAN sequence SOL 11. The ALTER sequence was proved comparing its results to default MSC/NASTRAN ones (see paragraph. 6). As expected, the results are exactly the same.

The paper is structured as follows. First, the mathematical background to obtain the exact solution of RMS responses to white noise excitation is reviewed.

Then, a direct application to structure-like systems is shown and the sequence of mathematical operations needed is identified. Afterwards, details of the implementation of this sequence into DMAP form is given, and the obtained results are validated through a representative example. Finally, details on possible extensions of the proposed method to more general problems is anticipated and discussed.

2 Mathematical Background

Ref. [1] gives a fairly complete and rigorous derivation of the theory of response of linear systems to white noise random excitation sources. Therefore, only a brief summary of the main results will be presented.

2.1 Output variance matrix for linear systems submitted to white noise

It is well known from Linear Systems Theory how to calculate the RMS response of a linear dynamic system submitted to white noise zero mean excitation. Such a system, may always be written in the form

$$\begin{aligned} \dot{x} &= Ax + Bw \\ z &= Cx \end{aligned} \tag{1}$$

where A is a stable matrix (i.e. all its eigenvalues have negative real parts)[1], x is the state vector, w is a vector of excitation sources consisting of zero mean white noise with PSD matrix W (symmetric and positive definite) and z is the system output. Note that the feedthrough term Dw is not present in the output vector definition. Otherwise, the RMS of the output would be infinite.

It can be demonstrated that the *steady state* variance matrix X of the state vector $x(t)$, defined as

$$X = \lim_{t\to\infty} \mathcal{E}\left(x(t)x(t)^T\right) \tag{2}$$

($\mathcal{E}$ is the mathematical expectance operator), is the solution of the linear Lyapunov equation

$$AX + XA^T + BWB^T = 0 \tag{3}$$

It can be proved that the above equation has a unique symmetric nonnegative-definite solution *if and only if* for all i and j the eigenvalues of the state matrix A verify the relation

[1]The response of an *unstable* system would be infinite.

$$\lambda_i + \lambda_j \neq 0 \tag{4}$$

This condition will always be met for structures having no rigid body modes. It is clear that the RMS displacements of a structure having rigid body modes will become infinite if the structure is submitted to white noise forces.

The steady state variance matrix, Z, of the output $z(t)$, i.e.

$$Z = \lim_{t\to\infty} \mathcal{E}\left(z(t)z(t)^T\right) \tag{5}$$

can be calculated from X by

$$Z = CXC^T \tag{6}$$

and finally, the steady state mean square values of the response are simply

$$\overline{z^2} = \mathrm{diag}\left(CXC^T\right) \tag{7}$$

Although the above Lyapunov equation might be rewritten as a linear system of equations in terms of the elements of the matrix X, it can be solved much more efficiently by using Schur decomposition techniques (see [3] for instance). However, since this is not possible within DMAP, we shall proceed in a different way.

3 Application to structural models

3.1 Dynamic equations of a structural system (frequency response problem).

The equilibrium equations of a linear dynamic structural problem is written in modal coordinates as follows

$$M_{hh}\ddot{u}_h + B_{hh}\dot{u}_h + K_{hh}u_h = P_{hp}f_p \tag{8}$$

The subindex h refers to the number of modes retained in the analysis. The matrix P_{hp} (where p is the number of independent load cases) is the matrix of modal forces; each column corresponds to the modal forces of a particular excitation case. Finally u_h is the vector of modal displacements. The load vector f_p is white noise zero mean process with PSD matrix W_{pp}.

The above equations can be written in the form of a first order linear system of differential equations. It takes the form

$$\dot{x}_e = A_{ee}x_e + P_{ep}f_p \tag{9}$$

where x_e is the state vector,

$$x_e = \left\{ \begin{array}{c} u_h \\ \dot{u}_h \end{array} \right\} \tag{10}$$

of order $2n_h$, A_{ee} is the system matrix,

$$A_{ee} = \left[\begin{array}{cc} 0_{hh} & I_{hh} \\ -M_{hh}^{-1} K_{hh} & -M_{hh}^{-1} B_{hh} \end{array} \right]$$

P_{ep} is the load influence matrix,

$$P_{ep} = \left[\begin{array}{c} 0_{hp} \\ M_{hh}^{-1} P_{hp} \end{array} \right]$$

In the case in which the normal modes are normalized with respect to the mass matrix, and for the case of modal structural damping, $M_{hh} = I_{hh}$, $K_{hh} = -\Omega_{hh}^2$ and $B_{hh} = G_{hh}\Omega_{hh}$ and the above matrices can be written

$$A_{ee} = \left[\begin{array}{cc} 0_{hh} & I_{hh} \\ -\Omega_{hh}^2 & -G_{hh}\Omega_{hh} \end{array} \right] \tag{11}$$

where 0_{hh} and I_{hh} are zero and identity matrices of order h respectively,

$$\Omega_{hh} = \operatorname{diag}\left(\omega_1, \omega_2, \ldots, \omega_h\right) \tag{12}$$

being ω_i the natural frequency of the i-th mode, and

$$G_{hh} = \operatorname{diag}\left(g_1, g_2, \ldots, g_h\right) \tag{13}$$

where g_i corresponds to the modal structural damping associated to mode i. Finally the load influence matrix P_{ep} is

$$P_{ep} = \left[\begin{array}{c} 0_{hp} \\ P_{hp} \end{array} \right] \tag{14}$$

3.2 Relation between modal coordinates and output variables

The response variables, that is those structural output variables in which the user is interested, can be expressed as a linear function of displacements, velocities and accelerations. Thus, they can be expressed in the form

$$z_k = \bar{C}_{kh}^0 u_h + \bar{C}_{kh}^1 \dot{u}_h + \bar{C}_{kh}^2 \ddot{u}_h$$

or, elliminating $\ddot{u}_h$ from (8)

$$\ddot{u}_h = M_{hh}^{-1}\left(P_{hp} f_p - B_{hh}\dot{u}_h - K_{hh} u_h\right) \tag{15}$$

$$
\begin{aligned}
z_k &= \bar{C}^0_{kh} u_h + \bar{C}^1_{kh} \dot{u}_h + \bar{C}^2_{kh} M^{-1}_{hh} \left(P_{hp} f_p - B_{hh} \dot{u}_h - K_{hh} u_h \right) \\
&= \left(\bar{C}^0_{kh} - \bar{C}^2_{kh} M^{-1}_{hh} K_{hh} \right) u_h + \left(\bar{C}^1_{kh} - \bar{C}^2_{kh} M^{-1}_{hh} B_{hh} \right) \dot{u}_h + \\
&\quad + \bar{C}^2_{kh} M^{-1}_{hh} P_{hp} f_p
\end{aligned}
\tag{16}
$$

which can be written as

$$
z_k = \left[\begin{array}{cc} C^0_{kh} & C^1_{kh} \end{array} \right] \left\{ \begin{array}{c} u_h \\ \dot{u}_h \end{array} \right\} + D_{kp} f_p = C_{ke} x_e + D_{kp} f_p
$$

where

$$
C_{ke} = \left[\begin{array}{cc} C^0_{kh} & C^1_{kh} \end{array} \right] \tag{17}
$$

and

$$
\begin{aligned}
C^0_{kh} &= \bar{C}^0_{kh} - \bar{C}^2_{kh} M^{-1}_{hh} K_{hh} \\
C^1_{kh} &= \bar{C}^1_{kh} - \bar{C}^2_{kh} M^{-1}_{hh} B_{hh} \\
D_{kp} &= \bar{C}^2_{kh} M^{-1}_{hh} B_{hp}
\end{aligned}
\tag{18}
$$

z_k is thus a vector of k elements that may represent any combination of all structural response variables (stresses, displacements, velocities, SPC forces, MPC forces, etc.)

In the problem we are dealing with, the matrix $\bar{C}^2_{kh}$ (and hence D_{kp}) must be zero, since otherwise the RMS of the response z_k would be infinite. Thus, the response vector is

$$
z_k = C_{ke} x_e
$$

Normally, the matrix C^1_{kh} will be zero, since the response variables requested by the user will be most likely displacements, stresses, element forces, etc. which do not depend on velocities. The submatrix C^0_{kh} can be recovered directly from MSC/NASTRAN as it will be seen later. However, the presence of non zero C^1_{kh} matrix does not alter the subsequent development.

The Lyapunov equation (3) can be solved explicitly as follows (see for instance [3]). Let Λ_{ee} and T_{ee} be the eigenvalues and eigenvectors matrices of the state matrix, i.e. they verify the equation

$$
A_{ee} T_{ee} = T_{ee} \Lambda_{ee} \tag{19}
$$

They are given explicitly by

$$
\Lambda_{ee} = \left[\begin{array}{cc} \Lambda_{hh} & 0_{hh} \\ 0_{hh} & \Lambda^H_{hh} \end{array} \right] \tag{20}
$$

and

$$
T_{ee} = \left[\begin{array}{cc} I_{hh} & I_{hh} \\ \Lambda_{hh} & \Lambda^H_{hh} \end{array} \right] \tag{21}
$$

the superindex H meaning complex conjugate transpose. The diagonal matrix Λ_{hh} can be easily calculated from natural frequencies and modal dampings as

$$\Lambda_{hh} = \Omega_{hh} \left(-\frac{1}{2} G_{hh} + j\sqrt{I_{hh} - \frac{1}{4} G_{hh}^2} \right) \tag{22}$$

where $j = \sqrt{-1}$. Using equation (19) the Lyapunov equation

$$A_{ee} X_{ee} + X_{ee} A_{ee}^T + P_{ep} W_{pp} P_{ep}^T = 0$$

can be written

$$T_{ee} \Lambda_{ee} T_{ee}^{-1} X_{ee} + X_{ee} \left(T_{ee} \Lambda_{ee} T_{ee}^{-1} \right)^T + P_{ep} W_{pp} P_{ep}^T = 0 \tag{23}$$

and letting

$$\begin{aligned} \bar{X}_{ee} &= T_{ee}^{-1} X_{ee} T_{ee}^{-H} \\ \bar{W}_{ee} &= T_{ee}^{-1} P_{ep} W_{pp} P_{ep}^T T_{ee}^{-H} \end{aligned}$$

equation (23) can be simplified to read

$$\Lambda_{ee} \bar{X}_{ee} + \bar{X}_{ee} \Lambda_{ee}^H + \bar{W}_{ee} = 0$$

Owing to the diagonal structure of Λ_{ee}, the matrix $\bar{X}_{ee}$ can be explicitly solved term by term,

$$\left(\bar{X}_{ee} \right)_{ij} = -\frac{\left(\bar{W}_{ee} \right)_{ij}}{\left(\Lambda_{ee} \right)_{ii} + \left(\Lambda_{ee}^H \right)_{jj}} \tag{24}$$

and the solution X_{ee} to the Lyapunov (23) equation is finally given by

$$X_{ee} = T_{ee} \bar{X}_{ee} T_{ee}^H \tag{25}$$

The steady state variance of the structural responses is

$$Z_{kk} = C_{ke} X_{ee} C_{ke}^T \tag{26}$$

and the steady state mean square values of the response,

$$\overline{z_k^2} = \operatorname{diag} \left(C_{ke} X_{ee} C_{ke}^T \right) \tag{27}$$

Finally, the steady state RMS value of the structural response is given by

$$z_{\mathrm{rms}} = \sqrt{\frac{\operatorname{diag} \left(C_{ke} X_{ee} C_{ke}^T \right)}{2}} \tag{28}$$

The factor 2 comes from the fact that in the above derivation it was supposed that the frequency spanned from $-\infty$ to $+\infty$, while in structural applications it is more usual to restrict the frequencies to take positive values only.

The expression (28) is the final results of our analysis. It gives the *exact* solution for the RMS values of the responses of a structure submitted to white noise of PSD given by the matrix W_{pp}.The whole procedure can now be summarized as follows:

1. Obtain the matrix of modal responses for the desired output variables. This will give the matrix C_{ke} or alternatively C^0_{ke} if output variables involve only modal displacements, which is by far more usual.

2. Obtain the eigenvalues and eigenvectors of the state matrix from the natural frequencies and modal dampings, by means of equations (22), (20) and (21)

3. Obtain the state variance matrix X_{ee} by means of equations (24) and (25)

4. Perform the matrix operation (28) to obtain the RMS values of the desired outputs.

4 Programming within MSC/NASTRAN

Following the operations sequence shown in preceding paragraph, a DMAP alter has been prepared for RANDOM problems solved within modal formulation. This formulation is more usual than the direct one when dealing with medium/high size structures. Particularly the alter is written for MSC/NASTRAN version 69, SOL 11.

The PSD matrix W_{pp} of the external forces must be introduced by the user via DMI cards on the Bulk Data Section.

With respect to the output variables, the sequence allows to recover RMS values of stresses, element forces and displacements. For the rest of output variables: SPC or MPC forces and so on, a few lines should be added to the alter. In particular, MPC forces can be calculated for instance using the theory given in [4].

The displacements are printed in the .f06 file in the standard format of static solutions of MSC/NASTRAN, while the stresses and element forces are printed in the .f06 file in matrix form (using module MATPRN), and an auxiliary index table is also printed to allow the identification of the different items (stresses or forces). It is not possible to print these results in the standard format of MSC/NASTRAN. This inconvenient arises from the poor operational capability of DMAP language when managing table datablocks (stresses and element forces are stored on tables while displacements are used both in matrix and table form).

However it is still possible to visualize the results in the same way as it is usually made in static analysis. Most postprocessors read stresses and element forces in table datablock format (as it is stored in the OUTPUT2 file). Therefore, a simple external computer program is needed to translate the Alter stress/force results (given in matrix form) into table datablock format. In the authors' case the postprocessor used was MSC/ARIES which eases considerably the above process since the input results file is written in ASCII format. Following this procedure, color plots, contour plots etc. of RMS displacements and stresses have been obtained without difficulty.

5 DMAP sequence

The sequence given in Appendix A is valid for MSC/NASTRAN version 69 SOL 11. It is valid for the obtention of RMS displacements, stresses and element forces. This requires the standard Case Control Commands DISP=ALL, etc. and the parameters OUTDIS, OUTSTR and OUTFOR (depending on the desired results) must be set to 'YES' in the Bulk Data section.

A parameter named NUMCO has been included aimed at the following purpose. The matrix Z_{kk} given in (26) must be calculated explicitly, although in fact only its main diagonal will be of use for us, as shown in equation (27). This matrix may be very large, so it is convenient to perform the matrix operation (26) by smaller blocks. If the user specifies some value for NUMCO, the triple matrix product will be performed in blocks of NUMCO rows, thus saving space and computation time. The parameter NUMCO has a default value of 400.

6 Examples

The DMAP Alter sequence has been validated, comparing its results with the outputs obtained from MSC/NASTRAN standard solution SOL 111. A relatively complex model (it represents an Optical monitoring camera, mounted onto the INTEGRAL ESA satellite) is used, and it is shown in Figure 1. The size of the model is about 10000 degrees of freedom, and about 1500 elements (most of them beam and shell types). The comparison for RMS displacements is shown in Figures 2 and 3. For stresses, the comparison is given in Table 1. The small differences arise from the fact of that MSC/NASTRAN uses numerical integration to derive the RMS values. Similar checkings have been done for element forces. The external input consists of two uncorrelated random excitations introduced as concentrated forces.



Figure 1: OMC Optical Monitoring Camera

D I S P L A C E M E N T   V E C T O R

POINT ID.	TYPE	T1	T2	T3	R1	R2	R3
350	G	1.877807E+00	1.572087E+00	1.079360E+00	1.244318E-02	1.499236E-02	1.844487E-02
6020	G	3.531184E+00	1.415994E+00	6.924799E-01	1.687280E-02	1.365854E-02	1.813710E-02
90010	G	8.147181E+00	6.145472E+00	1.417354E+00	9.573186E-02	6.429763E-02	1.196434E-01
90778	G	4.301331E+00	4.542489E+00	7.723124E-01	2.327813E-01	2.341773E-01	7.750175E-02

Figure 2. RMS displacements obtained by application of the method proposed in this paper

0 X Y - O U T P U T   S U M M A R Y (A U T O   O R   P S D F)

0	PLOT TYPE	CURVE TYPE	FRAME NO.	CURVE ID.	RMS VALUE	NO. POSITIVE CROSSINGS	XMIN FOR ALL DATA	XMAX FOR ALL DATA	YMIN FOR ALL DATA	X FOR YMIN	YMAX FOR ALL DATA	X FOR* YMAX
0	PSDF	DISP	0	350(3)	1.877518E+00	1.685858E+01	2.000E-01	6.020E+01	0.000E+00	5.910E+01	1.173E+00	1.440E+01
0	PSDF	DISP	0	350(4)	1.571824E+00	9.679505E+00	2.000E-01	6.020E+01	0.000E+00	5.910E+01	2.117E+00	6.700E+00
0	PSDF	DISP	0	350(5)	1.079080E+00	1.697989E+01	2.000E-01	6.020E+01	0.000E+00	5.910E+01	1.122E+00	7.300E+00
0	PSDF	DISP	0	350(6)	1.244140E-02	1.813201E+01	2.000E-01	6.020E+01	0.000E+00	5.910E+01	5.199E-05	7.300E+00
0	PSDF	DISP	0	350(7)	1.499149E-02	1.235287E+01	2.000E-01	6.020E+01	0.000E+00	5.910E+01	1.654E-04	6.700E+00
0	PSDF	DISP	0	350(8)	1.844023E-02	1.325200E+01	2.000E-01	6.020E+01	0.000E+00	5.910E+01	3.105E-04	6.700E+00
0	PSDF	DISP	0	6020(3)	3.526277E+00	1.898766E+01	2.000E-01	6.020E+01	0.000E+00	5.910E+01	3.082E+00	1.690E+01
0	PSDF	DISP	0	6020(4)	1.415804E+00	8.738486E+00	2.000E-01	6.020E+01	0.000E+00	5.910E+01	3.224E+00	6.700E+00
0	PSDF	DISP	0	6020(5)	6.919639E-01	1.027104E+01	2.000E-01	6.020E+01	0.000E+00	5.910E+01	6.816E-01	7.300E+00
0	PSDF	DISP	0	6020(6)	1.687180E-02	1.451736E+01	2.000E-01	6.020E+01	0.000E+00	5.910E+01	2.761E-04	1.410E+01
0	PSDF	DISP	0	6020(7)	1.365185E-02	2.294193E+01	2.000E-01	6.020E+01	0.000E+00	5.910E+01	1.094E-04	2.350E+01
0	PSDF	DISP	0	6020(8)	1.813249E-02	1.925008E+01	2.000E-01	6.020E+01	0.000E+00	5.910E+01	8.574E-05	1.940E+01
0	PSDF	DISP	0	90010(3)	8.147145E+00	1.790389E+01	2.000E-01	6.020E+01	0.000E+00	5.910E+01	4.674E+01	1.850E+01
0	PSDF	DISP	0	90010(4)	6.144311E+00	2.015849E+01	2.000E-01	6.020E+01	0.000E+00	5.910E+01	1.273E+01	1.700E+01
0	PSDF	DISP	0	90010(5)	1.416426E+00	1.922274E+01	2.000E-01	6.020E+01	0.000E+00	5.910E+01	7.599E-01	2.350E+01
0	PSDF	DISP	0	90010(6)	9.572046E-02	2.145818E+01	2.000E-01	6.020E+01	0.000E+00	5.910E+01	4.928E-03	1.950E+01
0	PSDF	DISP	0	90010(7)	6.429716E-02	1.833950E+01	2.000E-01	6.020E+01	0.000E+00	5.910E+01	3.404E-03	1.850E+01
0	PSDF	DISP	0	90010(8)	1.196380E-01	2.486071E+01	2.000E-01	6.020E+01	0.000E+00	5.910E+01	5.962E-03	2.350E+01
0	PSDF	DISP	0	90778(3)	4.301202E+00	1.894170E+01	2.000E-01	6.020E+01	0.000E+00	5.910E+01	6.118E+00	2.090E+01
0	PSDF	DISP	0	90778(4)	4.542355E+00	1.840977E+01	2.000E-01	6.020E+01	0.000E+00	5.910E+01	8.237E+00	1.960E+01
0	PSDF	DISP	0	90778(5)	7.722689E-01	1.545834E+01	2.000E-01	6.020E+01	0.000E+00	5.910E+01	4.792E-01	7.300E+00
0	PSDF	DISP	0	90778(6)	2.327814E-01	1.986796E+01	2.000E-01	6.020E+01	0.000E+00	5.910E+01	2.367E-02	1.960E+01
0	PSDF	DISP	0	90778(7)	2.341771E-01	1.989005E+01	2.000E-01	6.020E+01	0.000E+00	5.910E+01	2.403E-02	1.960E+01
0	PSDF	DISP	0	90778(8)	7.750043E-02	1.909940E+01	2.000E-01	6.020E+01	0.000E+00	5.910E+01	2.398E-03	1.950E+01

Figure 3. RMS displacements output from standard MSC/NASTRAN procedure

7 Future extensions

The capabilities of the procedure described in this paper can be substantially increased by further extensions. Description of these extensions that can be easily incorporated is given below.

- *Extension to direct formulation.* The procedure described as such, works only with modal formulation for dynamic problems. By far, this is the most usual approach. However, the algorithm can be easily reformulated in terms of M_{aa}, B_{aa} and K_{aa} (a meaning the degrees of freedom selected in ASET) instead of the corresponding modal matrices. Note however that the eigenvalues and eigenvectors of the corresponding system matrix A_{ee} must be calculated explicitly by using the CEAD module. Once Λ_{ee} and T_{ee} are calculated in this way, the rest of the procedure is the same.

- *Non diagonal matrices.* The user might be interested in including damping or stiffness matrices via DMIG cards. In this case, the mass, stiffness and/or damping matrices might not be diagonal. The consequences of this

RMS Stress Comparison (MPa)			
Element	Component	MSC/NASTRAN results	DMAP results
6510 (QUAD4)	$\sigma_x(z1)$	**87.26**	**87.28**
	$\sigma_y(z1)$	**68.78**	**68.79**
	$\tau_{xy}(z1)$	**20.69**	**20.69**
	$\sigma_x(z2)$	**52.39**	**52.40**
	$\sigma_y(z2)$	**81.92**	**81.93**
	$\tau_{xy}(z2)$	**24.27**	**24.27**
90005 (QUAD4)	$\sigma_x(z1)$	**294.27**	**294.27**
	$\sigma_y(z1)$	**89.62**	**89.62**
	$\tau_{xy}(z1)$	**46.56**	**46.57**
	$\sigma_x(z2)$	**300.88**	**300.88**
	$\sigma_y(z2)$	**128.20**	**128.20**
	$\tau_{xy}(z2)$	**82.13**	**82.14**
90220 (TRIA3)	$\sigma_x(z1)$	**403.10**	**403.12**
	$\sigma_y(z1)$	**954.96**	**954.98**
	$\tau_{xy}(z1)$	**173.03**	**173.10**
	$\sigma_x(z2)$	**274.14**	**274.18**
	$\sigma_y(z2)$	**385.30**	**385.31**
	$\tau_{xy}(z2)$	**137.22**	**137.30**

Table 1: RMS Stress Results Comparison

fact are twofold: first, the eigenvalues and eigenvectors of the resulting system matrix A_{ee} must be calculated explicitly, and second, the eigenvalues/vectors will not generally be arranged in the same order than that of the natural frequencies and modes of the structural model. Since the structural modal responses are ordered columnwise according to the structural modes, the user must be careful in ensuring that the order of system matrix eigenvectors is consistent.

- *Non uniform spectrum.* The results described in this paper are only valid for zero mean white noise excitation forces. Extensions to more general cases of excitation is possible and easy. Most of the more usual shapes of input spectra in the aerospace field can be tackled with minor modifications of the algorithm just described. However, due to the lack of space, this procedure will be described in detail in a future paper.

8 Conclusions

An efficient method for computation *exact* RMS values of structural response variables of structures submitted to white noise excitation forces has been described and can be easily appended to the MSC/NASTRAN Frequency Response solution. A DMAP Alter has been prepared and described to that purpose. The method allows the calculation of RMS values of displacements, stresses, element forces and in general all conceivable output variables that can be expressed as linear functions of displacements and velocities. Results have been compared to those provided by MSC/NASTRAN standard method, and have been proved to be the same in all cases.

Main advantages of the described method are

- It allows the postprocessing of RMS results exactly in the same manner as if they were obtained from a conventional static analysis (i.e., color and contour plots over the whole structure)
- The procedure provides exact solutions so that piecewise numerical integration of the PSD of the response is not needed. Thus the procedure is by far much more efficient than the standard one implemented in MSC/NASTRAN.
- Owing to this fact, there is no limitation on the number of structural response variables requested by the user, as it happens in fact with the standard procedure.
- Extensions of the capabilities of this procedure to more general problems are easy and require only minor modifications.

9 References

[1] H. Kwakernaak and R. Sivan. *Linear Optimal Control Systems*, John Wiley and Sons, Inc., New York, 1972.
[2] J.L. Junkins and Y. Kim. *Introduction to the Dynamics and Control of Flexible Structures.* AIAA, 1992.
[3] Anonymous,MATLAB *Control Systems Toolbox*, User's Guide. The Mathworks Inc.
[4] E. Fuente and J. San Millán. "Calculation within MSC/NASTRAN of the forces transmitted by the Multipoint Constraints (MPC) and the Forces Generated in Suport Constraints". MSC World Users Conference, Newport Beach (CA), June 5th, 1996.

10 Appendix A. DMAP Sequence

```
$.............................................................
$
$ DMAP SEQUENCE FOR THE OBTENTION OF RMS VALUES OF
$ DISPLACEMENTS, ACCELERATIONS, STRESSES AND ELEMENT FORCES,
$ IN A WHITE NOISE RANDOM ANALYSIS
$
$
$ THIS DMAP ALTER IS VALID FOR:
$
$ *** MSC/NASTRAN VERSION 69 Y SOL 11 (MFREQ) ******
$ ————————————————————————————————————————
$
$
$ THE OUTPUT IS OBTAINED USING THE STANDARD
$ CASE CONTROL INSTRUCTIONS:
$
$ DISP = ALL
$ STRESS = ALL
$ FORCES = ALL
$
$
$ MOREOVER IT IS NECESSARY TO INTRODUCE THE FOLLOWING
$ PARAMETERS INTO THE BULK DATA SECTION:
$
$
$ PARAM, OUTDIS, YES (FOR OBTENTION OF DISPLACEMENTS)
$ PARAM, OUTSTR, YES (STRESSES)
$ PARAM, OUTFOR, YES (ELEMENT FORCES)
$
$
$
$ PARAMETER NUMCO: INTEGER NUMBER NECESSARY FOR HARD DISK
$ OPTIMIZATION. ITS DEFAULT VALUE IS 400, BUT MAY BE CHANGED
$ VIA BULK DATA (RECOMMENDED RANGE IS 50 - 1000)
$
$————————————————————————————————————————————-
$
$ DMAP ALTER ALTERRANDOM.V69
$ ——————————————————
$
$
$ PREPARED BY F. J. SAN MILLAN
$ DESIGNED BY E. DE LA FUENTE
$
$
$ AUGUST 1997
$
$ ——————————————————————————————————————-
$ I.N.T.A. NATIONAL INSTITUTE FOR AEROSPACE RESEARCH
$ STRUCTURES AND MECHANISMS DEPARTMENT
$ ——————————————————————————————————————-
$
$——————————————————————————
COMPILE SOL11 SOUIN = MSCSOU LIST NOREF
ALTER 6 $
$
TYPE PARM,,I,, II, JJ $
TYPE PARM,,I,, NUMFR, NCOLPH, NPP, NH, NE $
TYPE PARM,,I,, COPH, N1, NTOTAL, NRESID $
TYPE PARM,,CHAR3,Y,OUTDIS='NO' $
TYPE PARM,,CHAR3,Y,OUTSTR='NO' $
TYPE PARM,,CHAR3,Y,OUTFOR='NO' $
TYPE PARM,,I,Y,NUMCO=400 $
$.....................................
$
$ *** SPP = PSD MATRIX OF EXTERNAL LOADS (NPP * NPP) ***
$
$ NPP = NUMBER OF LOAD CASES
$
$.....................................
ALTER 497 $
DMIIN DMI,DMINDX/SPP, ANN,BNP,CPN, , , , , , / $
DTIIN DTI,DTINDX/CONTROL,,,,,,,,, $
PARAML SPP//'TRAILER'/1/S,N,NPP $
$......................................
$
$ **** OBTENTION OF MODAL MATRICES ****
$
$ MESS = MODAL STRESSES MATRIX
$ MEFF = MODAL FORCES MATRIX
$
$.......................................
ALTER 612, 612 $
ALTER 619 $
IF (OUTSTR='YES' OR OUTFOR='YES') THEN $
```

```
DRMS1 IPHIG1,IQG1,IES1,IEF1/TPHHIP,MPHHIP,TQQP,MQQP,TESS,
MESS,TEFF,MEFF $
ELSE $
ENDIF $
$..........................................
$
$ *** OBTENTION OF MATRIX PHP (NH * NPP) ***
$ THIS MATRIX CONTAINS THE MODAL FORCES FOR THE
$ DIFFERENT LOAD CASES.
$
$ NH = NUMBER OF MODES COMPUTED IN THE PROBLEM
$ NPP = NUMBER OF LOAD CASES
$..........................................
ALTER 639 $
NH = NEIGV $
PARAML PH//'TRAILER'/1/S,N,NCOLPH $
NUMFR = NCOLPH/NPP $
$
II = 0 $
FILE PHP=APPEND $
$
DO WHILE (II<NPP) $
COPH = 1 + II*NUMFR $
MATMOD PH,,,,,/PH1C,/1/COPH $
APPEND PH1C,/PHP/2 $
II = II + 1 $
ENDDO
$...............................
$
$ OBTENTION OF MATRIX BEP ( NE * NPP )
$
$ THE FIRST NH ROWS OF THE MATRIX ARE NULL
$ IN A GENERAL PROBLEM
$
$ NE = 2 * NH
$............................
NE = NH * 2 $
MATGEN ,/ZEROHP/7/NH/NPP $
MATGEN ,/RP1/6/NE/NH/NH $
MERGE ZEROHP,PHP,,,,RP1/BEP/1 $
$..........................................
$
$ ** OBTENTION OF MATRIX WEE (NE * NE) **
$
$
$ WEE = BEP * SPP* (BEP)T
$..........................................
TRNSP BEP/BEPT $
SMPYAD BEP,SPP,BEPT,,,/WEE/3/1 $
DELETE /PH1C,BEP,BEPT, ,/ $
$..................................
$
$ ***** OBTENTION OF COMPLEX EIGENVALUES MATRIX ******
$
$ EIGVAMT ( NE * NE )
$
$ AND:
$
$ COMPLEX EIGENVECTORS MATRIX
$
$ TEE ( NE * NE )
$
$
$ DEFINITION OF MATRICES:
$
$ - OMHH ( NH * NH ): DIAGONAL MATRIX CONTAINING NATURAL
$ FREQUENCIES OF STRUCTURE
$
$ - CHIHH ( NH * NH ): DIAGONAL MATRIX CONTAINING MODAL
$ DAMPING VALUES
$.....................................
DIAGONAL KHH/OMHH/'WHOLE'/0.5 $
MATGEN ,/IHH/1/NH $
MATGEN ,/NUHH/7/NH/NH $
$
ADD BHH,OMHH/CHIHH//2.0/2 $
DIAGONAL CHIHH/CHIHH2/'WHOLE'/2.0 $
ADD IHH,CHIHH2/IMAGHH2//-1./0 $
DIAGONAL IMAGHH2/IMAGHH/'WHOLE'/0.5 $
$
ADD OMHH,CHIHH/REALHH/-1.//1 $
ADD OMHH,IMAGHH/IMHH///1 $
$
ADD REALHH,IMHH/EGVT/1./(0.,1.)/0 $
ADD REALHH,IMHH/EGVTCJ/1./(0.,-1.)/0 $
$
MATGEN ,/CRPP/6/NE/NH/NH $
MERGE EGVT,NUHH, NUHH,EGVTCJ, CRPP,/EIGVAMT $
```

```
MERGE IHH,EGVT,IHH,EGVTCJ,CRPP,/TEE///1 $
$.................................................
$
$ **** OBTENTION OF MATRIX XEE ****
$ ( NE * NE)
$
$...............................................
DIAGONAL EIGVAMT/EGVRE/'COLUMN' $
ADD EIGVAMT,/EIGVABIS/(0.,-1.) $
DIAGONAL EIGVABIS/EGVIM/'COLUMN' $
ADD EGVRE,EGVIM/EGVEE//(0.,1.)/0 $
$
FILE XIIMAT =APPEND $
II = 1 $
DO WHILE (II<=NE) $
APPEND EGVEE,/XIIMAT/2 $
II = II + 1 $
ENDDO $
$
TRNSP XIIMAT/XJJMAT $
MATMOD XJJMAT,,,,,/XJJMATH,/10 $
ADD XIIMAT,XJJMATH/XEE $
$..........................................
$
$ OBTENTION OF MATRIX WEEB (NE * NE)
$
$ WEEB = TEEINV * WEE * (TEEINV)H
$
$ INDEX H MEANS TRANSPOSED COMPLEX CONJUGATED
$ ————————————————————————————————————————————
$...........................................
SOLVE TEE,/TEEINV/3 $
TRNSP TEEINV/TEEINVT $
MATMOD TEEINVT,,,,,/TEEINVH,/10 $
SMPYAD TEEINV,WEE,TEEINVH,,,/WEEB/3/1 $
$.........................................
$
$ OBTENTION OF MATRIX PEEB (NE * NE)
$
$ PEEB(I,J) = - WEEB(I,J)/XEE(I,J)
$
$........................................
ADD WEEB,XEE/PEEB/(-1.,0.)//2 $
$........................................
$
$ OBTENTION OF MATRIX PEE (NE * NE)
$
$ PEE = TEE * PEEB * (TEE)H
$
$ INDEX H MEANS TRANSPOSED COMPLEX CONJUGATED
$ ————————————————————————————————————————————
$.........................................
TRNSP TEE/TEET $
MATMOD TEET,,,,,/TEEH,/10 $
SMPYAD TEE,PEEB,TEEH,,,/PEE/3/1 $
$MATPRN PEE// $
$.............................................
$
$ *** CALCULATION OF DISPLACEMENT MATRIX (URMS) ***
$ ———————————————————————————————————————————————
$
$ THIS MATRIX CONTAINS ALL DISPLACEMENTS RMS VALUES
$
$ URMS IS A MATRIX (NDOF * 1)
$ NDOF = NUMBER OF DEGREES OF FREEDOM OF THE STRUCTURE
$
$.............................................
MATGEN ,/UU1/6/NE/NH/NH $
MATGEN ,/IDMAT1/1/1/1 $
FILE MAT1TR=SAVE $
ADD IDMAT1,/MAT1TR/2.0E-60 $
$
$
IF (OUTDIS = 'YES') THEN $
PARAML PHIGH//'TRAILER'/2/S,N,NDOF $
$
MATGEN ,/NUGH/7/NDOF/NH $
MERGE PHIGH,,NUGH,,UU1,/PHIEH/1 $
$
N1 = INT(NDOF/NUMCO) $
$
II = 1 $
FILE U4RMS=APPEND $
DO WHILE (II <=N1) $
NTOTAL = NDOF - NUMCO * (II-1) $
NRESID = NDOF - NUMCO*II $
MATGEN ,/UU3/6/NTOTAL/NUMCO/NRESID $
PARTN PHIEH,,UU3/PHIE1,PHIE2,,/1 $
```

```
TRNSP PHIE1/PHIE1T $
$
SMPYAD PHIE1,PEE,PHIE1T,,,/U1RMS/3 $
DIAGONAL U1RMS/U2RMS/'COLUMN' $
TRNSP U2RMS/U3RMS $
$
APPEND U3RMS,/U4RMS/2 $
EQUIVX PHIE2/PHIEH/ALWAYS $
II = II + 1 $
ENDDO $
$
DELETE /PHIGH,PHIE1,PHIE1T,U1RMS,U2RMS/ $
DELETE /U3RMS,UU3,,,/ $
$
TRNSP PHIEH/PHIEHT $
$
SMPYAD PHIEH,PEE,PHIEHT,,,/U1RMS/3 $
DIAGONAL U1RMS/U2RMS/'COLUMN' $
TRNSP U2RMS/U3RMS $
APPEND U3RMS,/U4RMS/2 $
$
$
ADD U4RMS,/U5RMS/0.5 $
TRNSP U5RMS/U6RMS $
DIAGONAL U6RMS/URMS/'WHOLE'/0.5 $
$
DELETE /U4RMS,U5RMS,U6RMS,PHIEH,PHIEHT/ $
$.........................................
$
$ **** FORMATTED OUTPUT FOR DISPLACEMENTS ****
$ ——————————————————————————————-
$
$ OUGRMS = DISPLACEMENTS
$
$.........................................
SDR2 CASECC,CSTM,MPT,DIT,EQEXIN,,ETT,EDT,BGPDT,,,URMS,EST,
XYCDB,,,/,,OUGRMS,,,/'STATICS'/S,N,NOSORT2 $
OFP OUGRMS//S,N,CARDNO/1 $
DELETE /PHIGHT,URMS2,URMS2T,URMSS,URMS/ $
ELSE $
ENDIF $
$...............................................
$
$ *** CALCULATION OF ELEMENT STRESS MATRIX (OESRMS) ***
$ ————————————————————————————————————
$
$ THIS MATRIX CONTAINS ALL STRESS RMS VALUES
$
$ OESRMS IS A MATRIX (NSTCO * 1)
$ NSTCO = NUMBER OF STRESS COMPONENTS OF STRUCTURE
$
$ THE OUTPUT IS:
$
$ - OESRMS : RMS STRESS VALUES
$
$ - TESSN : INDEX TABLE FOR DISTINCTION OF THE DIFFERENT
$ STRESS COMPONENTS WITHIN MATRIX OESRMS
$...........................................................................................................
IF (OUTSTR = 'YES') THEN $
PARAML MESS//'TRAILER'/2/S,N,NSTCO $
$
MATGEN ,/NUSTH/7/NSTCO/NH $
MERGE MESS,,NUSTH,,UU1,/MESSE/1 $
$
N1 = INT(NSTCO/NUMCO) $
$
II = 1 $
FILE ES4RMS=APPEND $
DO WHILE (II<=N1) $
NTOTAL = NSTCO - NUMCO * (II-1) $
NRESID = NSTCO - NUMCO*II $
MATGEN ,/UU4/6/NTOTAL/NUMCO/NRESID $
PARTN MESSE,,UU4/MESS1,MESS2,,/1 $
TRNSP MESS1/MESS1T $
$
SMPYAD MESS1,PEE,MESS1T,,,/ES1RMS/3 $
DIAGONAL ES1RMS/ES2RMS/'COLUMN' $
TRNSP ES2RMS/ES3RMS $
$
APPEND ES3RMS,/ES4RMS/2 $
EQUIVX MESS2/MESSE/ALWAYS $
II = II + 1 $
ENDDO $
$
DELETE /MESS, MESS1,MESS1T,ES1RMS,ES2RMS/ $
DELETE /ES3RMS,UU4,,,/ $
$
TRNSP MESSE/MESSET $
```

```
$
SMPYAD MESSE,PEE,MESSET,,,/ES1RMS/3 $
DIAGONAL ES1RMS/ES2RMS/'COLUMN' $
TRNSP ES2RMS/ES3RMS $
APPEND ES3RMS,/ES4RMS/2 $
$
$
ADD ES4RMS,/ES5RMS/0.5 $
TRNSP ES5RMS/ES6RMS $
DIAGONAL ES6RMS/OESRMS/'WHOLE'/0.5 $
$
TABEDIT TESS,CONTROL,,,/TESSN $
MATPRN TESSN, OESRMS// $
OUTPUT4 OESRMS,,,,//-1/11/2 $
DELETE /ES4RMS,ES5RMS,ES6RMS,MESSE,MESSET/ $
ELSE $
ENDIF $
$..........................................................................................
$
$ *** CALCULATION OF ELEMENT FORCE MATRIX (OEFRMS) ***
$ ————————————————————————————————————————-
$
$ THIS MATRIX CONTAINS ALL ELFORCE RMS VALUES
$
$ OEFRMS IS A MATRIX (NFOCO * 1)
$ NFOCO = NUMBER OF ELFORCE COMPONENTS OF STRUCTURE
$
$ THE OUTPUT IS:
$
$ - OEFRMS : RMS FORCE VALUES
$
$ - TEFFN : INDEX TABLE FOR DISTINCTION OF THE DIFFERENT
$ FORCE COMPONENTS WITHIN MATRIX OEFRMS
$....................................
IF (OUTFOR = 'YES') THEN $
PARAML MEFF//'TRAILER'/2/S,N,NFOCO $
$
MATGEN ,/NUFOH/7/NFOCO/NH $
MERGE MEFF,,NUFOH,,UU1,/MEFFE/1 $
$
N1 = INT(NFOCO/NUMCO) $
$
II = 1 $
FILE EF4RMS=APPEND $
DO WHILE (II<=N1) $
NTOTAL = NFOCO - NUMCO*(II-1) $
NRESID = NFOCO - NUMCO*II $
MATGEN ,/UU5/6/NTOTAL/NUMCO/NRESID $
PARTN MEFFE,,UU5/MEFF1,MEFF2,,/1 $
TRNSP MEFF1/MEFF1T $
$
SMPYAD MEFF1,PEE,MEFF1T,,,/EF1RMS/3 $
DIAGONAL EF1RMS/EF2RMS/'COLUMN' $
TRNSP EF2RMS/EF3RMS $
$
APPEND EF3RMS,/EF4RMS/2 $
EQUIVX MEFF2/MEFFE/ALWAYS $
II = II + 1 $
ENDDO $
$
DELETE /MEFF, MEFF1,MEFF1T,EF1RMS,EF2RMS/ $
DELETE /EF3RMS,UU5,,,/ $
$
TRNSP MEFFE/MEFFET $
$
SMPYAD MEFFE,PEE,MEFFET,,,/EF1RMS/3 $
DIAGONAL EF1RMS/EF2RMS/'COLUMN' $
TRNSP EF2RMS/EF3RMS $
APPEND EF3RMS,/EF4RMS/2 $
$
$
ADD EF4RMS,/EF5RMS/0.5 $
TRNSP EF5RMS/EF6RMS $
DIAGONAL EF6RMS/OEFRMS/'WHOLE'/0.5 $
$
TABEDIT TEFF,CONTROL,,,/TEFFN $
MATPRN TEFFN, OEFRMS// $
OUTPUT4 OEFRMS,,,,//-1/13/2 $
DELETE /EF4RMS,EF5RMS,EF6RMS,MEFFE,MEFFET/ $
ELSE $
ENDIF $
$.........................................
$
$ PREVENT STANDARD OUTPUT OF SOL 30
$
$.......................................
ALTER 810,810 $
ALTER 828,828 $
```

```
$..............................................................
$
$ INSERT INTO THE BULK DATA SECTION
$
$..............................................................
$DTI CONTROL 1 DR 1 2
$
$..............................................................................
$
$ INSERT INTO THE FMS SECTION
$ ——————————————————————
$..............................................................................
$ASSIGN OUTPUT4='C:\users\millanfj\NASTRAN.proofs\stress.ou4' UNIT=11,
$ FORM=FORMATTED DELETE
$ASSIGN OUTPUT4='C:\users\millanfj\NASTRAN.proofs\force.ou4' UNIT=13,
$ FORM=FORMATTED DELETE
```